Na tym wykładzie chciałbym przekonać Państwa, że *Mathematica*® może być pomocna w studiowaniu analizy matematycznej.

Liczby i działania na liczbach

Zmierzamy w stronę badania przebiegu zmienności funkcji rzeczywistej jednej zmiennej, ale po drodze powinniśmy poznać kilka zagadnień takich jak liczenie granic funkcji, znajdowanie dziedziny i przeciwdziedziny, liczenie pochodnych, znajdowanie miejsc zerowych pierwszej i drugiej pochodnej. Przydatne będą też instrukcje pozwalające na narysowanie kompletnego wykresu funkcji (łącznie z asymptotami).
Do podstawowych umiejętności należy jednak określanie typów liczb. Jest to niezbędne w szczególności przy upraszczaniu wyrażeń, więc od tego zaczniemy.

```
In[1]:= (* Zacznijmy od zadania pytania, na które znamy odpowiedź *)

In[2]:= 2 + 2 == 4

Out[2]= True

In[3]:= (* Dokładnie w ten sam sposób możemy
         sprawdzać przynależność do określonego typu liczb *)

In[4]:= (* liczby całkowite *)

In[5]:= Element[-1, Integers]
        należy do   zbiór liczb ca

Out[5]= True

In[6]:= (* inaczej, używając pomocy Writing Assistant  *)

In[7]:= -1 ∈ ℤ

Out[7]= True

In[8]:= (* Dlaczego takie sprawdzenie jest ważne ? *)

In[9]:= Simplify[Sin[n * Pi]]
        uprość   sinus  pi

Out[9]= Sin[n π]

In[10]:= (* Tego wyrażenia w ogólnym przypadku nie da się uprościć,
         ale dla n całkowitego ... *)

In[11]:= Simplify[Sin[n * Pi], Element[n, Integers]]
         uprość   sinus  pi    należy do  zbiór liczb cał

Out[11]=
         0

In[12]:= (* ... przyjmuje ono prostą wartość. *)

In[13]:= (* liczby pierwsze *)
```

```
In[14]:= Element[5, Primes]
         należy do  zbiór liczb p
Out[14]=
         True

In[15]:= Element[15, Primes]
         należy do    zbiór liczb
Out[15]=
         False

In[16]:= (* dzielniki liczby calkowitej *)

In[17]:= Divisors[17]
         dzielniki
Out[17]=
         {1, 17}

In[18]:= Divisors[36]
         dzielniki
Out[18]=
         {1, 2, 3, 4, 6, 9, 12, 18, 36}

In[19]:= (* największy wspólny podzielnik (greatest common divisor) *)

In[20]:= GCD[30, 15, 100]
         największy wspólny dzielnik
Out[20]=
         5

In[21]:= (* najmniejsza wspólna wielokrotność (least common multiple) *)

In[22]:= LCM[12, 50, 8]
         najmniejsza wspólna wielokrotność
Out[22]=
         600

In[23]:= (* liczby wymierne *)

In[24]:= Element[3/5, Integers]
         należy do    zbiór liczb ca
Out[24]=
         False

In[25]:= Element[3/5, Rationals]
         należy do    liczby wymiern
Out[25]=
         True

In[26]:= Element[Pi, Rationals]
         należy do pi liczby wymiern
Out[26]=
         False
```

```
In[27]:= Element[E, Rationals]
         należy do |l··· |liczby wymiern
Out[27]=
         False

In[28]:= (* liczby rzeczywiste *)

In[29]:= Element[E, Reals]
         należy do |l··· |liczby rz
Out[29]=
         True

In[30]:= Element[Pi, Reals]
         należy do |pi   |liczby rz
Out[30]=
         True

In[31]:= (* Liczby zespolone *)

In[32]:= (* słynna liczba i: podniesiona do kwadratu daje liczbę -1 *)

In[33]:= I
         jedność urojona
Out[33]=
         i

In[34]:= I*I
         ··· |jedność urojona
Out[34]=
         -1

In[35]:= Element[I, Reals]
         należy do |··· |liczby rz
Out[35]=
         False

In[36]:= Element[I, Complexes]
         należy do |··· |liczby zespolor
Out[36]=
         True

In[37]:= z = I; z^2 + 1 == 0
             jedność urojona
Out[37]=
         True

In[38]:= z =.; (* z staje się z powrotem nieokreślone *)

In[39]:= Element[3 - 4*I, Complexes]
         należy do     |··· |liczby zespolor
Out[39]=
         True

In[40]:= (* Upraszczanie wyrażeń *)

In[41]:= (* W tym przypadku Mathematica zrobi wszystko "bez popychania", ... *)
```

```
In[42]:= ((2/5)/(2+1/2))*((4+1/5)-(1+3/40))+(135/100)/(27/10)
Out[42]=
1

In[43]:= (* ... ale w wielu przypadkach trzeba pomóc, dostarczając dodatkowe informacje *)

In[44]:= Simplify[Abs[x] + Abs[x + 1] + Sqrt[x^2 + 4*x + 4], Element[x, Reals]]
          uprość   wartoś…  wartość b…   pierwiastek kwadratowy   należy do   liczby rze
Out[44]=
Abs[x] + Abs[1 + x] + Abs[2 + x]

In[45]:= (* Uproszczamy wyrażenia z logarytmami *)

In[46]:= ? Log
           logarytm
Out[46]=
```

Symbol

Log[z] gives the natural logarithm of z (logarithm to base e).

Log[b, z] gives the logarithm to base b.

```
In[47]:= Log[E]
          l…  liczba Eulera
Out[47]=
1

In[48]:= Log[Sqrt[E]]
          l…  pie…  licz
Out[48]=
```

$$\frac{1}{2}$$

```
In[49]:= (* bo ... *)

In[50]:= E^(1/2) == Sqrt[E]
          liczba Eul…  pie…  lic:
Out[50]=
True

In[51]:= (* Dwa kompletnie równoważne zapisy logarytmu dziesiętnego;
         stąd aż TRZY znaki równości *)

In[52]:= Log10[z] === Log[10, z]
          logarytm dz…  logarytm
Out[52]=
True

In[53]:= Log10[1000]
          logarytm dziesiętny
Out[53]=
3

In[54]:= (* bo ... *)
```

In[55]:= `10^3 == 1000`

Out[55]= `True`

In[56]:= `Log[64, 1/4]`
logarytm

Out[56]= $-\frac{1}{3}$

In[57]:= `(* bo ... *)`

In[58]:= `64^(-1/3) == 1/4`

Out[58]= `True`

In[59]:= `Log[32, 1/2]`
logarytm

Out[59]= $-\frac{1}{5}$

In[60]:= `Log[4, Sqrt[2]/2]`
logar··· pierwiastek k

Out[60]= $-\frac{\text{Log}[2]}{2\ \text{Log}[4]}$

In[61]:= `(* Powyższy wynik nie wygląda zbyt prosto, więc próbujemy Simplify, ... *)`
uprość

In[62]:= `Simplify[-Log[2]/(2 Log[4])]`
uprość

Out[62]= $-\frac{\text{Log}[2]}{\text{Log}[16]}$

In[63]:= `(* ... ale otrzymany wynik też jest skomplikowany. Dopiero FullSimplify załatwia sprawę ! *)`
uprość pełniej

In[64]:= `FullSimplify[-Log[2]/Log[16]]`
uprość pełniej

Out[64]= $-\frac{1}{4}$

In[65]:= `(* Podobnie jest dla wyrażenia 36^Log[6,5] *)`
logarytm

In[66]:=
```
36 ^ Log[6, 5]
```
logarytm

Out[66]=

$36^{\frac{Log[5]}{Log[6]}}$

In[67]:=
```
Simplify[36 ^ Log[6, 5]]
```
uprość logarytm

Out[67]=

$36^{\frac{Log[5]}{Log[6]}}$

In[68]:=
```
FullSimplify[36 ^ Log[6, 5]]
```
uprość pełniej logarytm

Out[68]=

25

In[69]:=
```
(* Niestety, często próba użycia FullSimplify prowadzi do bardzo długich,
wręcz niekończących się obliczeń. Dlatego
 tej instrukcji należy używać rozważnie ! *)
```
uprość pełniej

In[70]:=
```
(* Rozkład na ułamki proste *)
```

In[71]:=

$\text{Apart}\left[\frac{a+2\,a\,b+b^2}{a^2-b^2}\right]$

rozłóż na ułamki proste

Out[71]=

$$-1+\frac{-1-3\,a}{2\,(-a+b)}+\frac{1-a}{2\,(a+b)}$$

In[72]:=
```
(* Rozłożyć na czynniki *)
```

In[73]:=
```
Factor[x ^ 6 + 6 * x ^ 5 - x ^ 4 - 6 * x ^ 3]
```
rozłóż na czynniki

Out[73]=

$(-1+x)\,x^3\,(1+x)\,(6+x)$

In[74]:=
```
(* reszta z dzielenia wielomianów *)
```

In[75]:=
```
PolynomialRemainder[x ^ 5 + 3 * x ^ 4 + 2 * x ^ 3 - 5 * x ^ 2 + 3 * x + 7, x ^ 2 + 2, x]
```
reszta z dzielenia wielomianów

Out[75]=

$29+3\,x$

In[76]:=
```
(* Dla jakiej wartosci k wielomian W(x)=
 x^5+3*x^4+2*x^3-5*x^2+3*x+k jest podzielny przez (x-2) ?
    Inaczej mówiąc, dla jakiej wartości k reszta
 z dzielenia wielomianu W(x) przez (x-2) jest równa zero ? *)
```

In[77]:=
```
PolynomialRemainder[x ^ 5 + 3 * x ^ 4 + 2 * x ^ 3 - 5 * x ^ 2 + 3 * x + k, x - 2, x]
```
reszta z dzielenia wielomianów

Out[77]=

$82+k$

In[78]:=
```
(* Aby ta reszta była równa zero, k musi być równe -82 ! *)
```

In[79]:=
```
(* sprawdzenie *)
```

In[80]:=
```
Simplify[(x^5 + 3*x^4 + 2*x^3 - 5*x^2 + 3*x - 82)/(x - 2)]
uprość
```

Out[80]=

$41 + 19 x + 12 x^2 + 5 x^3 + x^4$

In[81]:=
```
(* Niekiedy mamy informację o pewnej zmiennej
 i mamy ją wykorzystać do policzenia innego wyrażenia,
w którym ta zmienna występuje. Przykład: wiemy, że x^2 +2/x^2 = 5,
a mamy policzyć x^4 + 4/x^4 oraz  Sqrt[2]/x-x. Jest to typowy problem,
                                  pierwiastek kwadratowy
w którym wykorzystujemy wzory skróconego mnożenia,
ale mając do dyspozycji program Mathematica,
możemy wykorzystać funkcję Eliminate *)
                           wyeliminuj zmienne
```

In[82]:=
```
Eliminate[{x^2 + 2/x^2 == 5, g1 == x^4 + 4/x^4}, x]
wyeliminuj zmienne
```

Out[82]=

$g1 == 21$

In[83]:=
```
Eliminate[{x^2 + 2/x^2 == 5, g2 == Sqrt[2]/x - x}, x]
wyeliminuj zmienne                   pierwiastek kwadratow
```

Out[83]=

$g2^2 == 5 - 2 \sqrt{2}$

In[84]:=
```
(* Podobne zadanie, w którym występują dwa warunki: Uzasadnij,
że jeżeli a+b=1 i a^2+b^2=7, to a^4+b^4==31 *)
```

In[85]:=
```
Eliminate[{a + b == 1, a^2 + b^2 == 7, g3 == a^4 + b^4}, {a, b}]
wyeliminuj zmienne
```

Out[85]=

$-31 + g3 == 0$

Równania i nierówności

In[86]:=
```
(* BARDZO WAŻNE: podstawienie ! *)
```

In[87]:=
```
y /. {y → 450}
```

Out[87]=

450

In[88]:=
```
(y^2 + 1) /. {y → 450}
```

Out[88]=

202 501

In[89]:=
```
y
```

Out[89]=

y

In[90]:=
```
x /. {x → 450}
```
Out[90]=
```
450
```
In[91]:=
```
x
```
Out[91]=
```
x
```
In[92]:=
```
(* Samo podstawienie nie zmienia wartości x *)
```
In[93]:=
```
(* Uwaga: Solve nie nadaje bezpośrednioo wartości pierwiastkom równania, ... *)
```
rozwiąż równanie

In[94]:=
```
Solve[x^3 - x^2 + 2 x + 4 == 0, x]
```
rozwiąż równanie

Out[94]=

$\{\{x \to -1\}, \{x \to 1 - i \sqrt{3}\}, \{x \to 1 + i \sqrt{3}\}\}$

In[95]:=
```
(* ... ale można te wartości uzyskać w następujący sposób: *)
```
In[96]:=
```
S1 = Solve[x^3 - x^2 + 2 x + 4 == 0, x]
```
rozwiąż równanie

Out[96]=

$\{\{x \to -1\}, \{x \to 1 - i \sqrt{3}\}, \{x \to 1 + i \sqrt{3}\}\}$

In[97]:=
```
S1[[1]]
```
Out[97]=
```
{x → -1}
```
In[98]:=
```
(* S1[[1]] jest już gotowym elementem instrukcji podstawienia;
podobnie S1[[2]] i S1[[3]] *)
```
In[99]:=
```
x1 = x /. S1[[1]]
```
Out[99]=
```
-1
```
In[100]:=
```
x2 = x /. S1[[2]]
```
Out[100]=

$1 - i \sqrt{3}$

In[101]:=
```
x3 = x /. S1[[3]]
```
Out[101]=

$1 + i \sqrt{3}$

In[102]:=
```
ClearAll[S1]; (* w ten sposób S1 staje się nieokreślone *)
```
wyczyść wszystko

In[103]:=
```
(* Powtarzam:
 po wykonaniu Solve sama wielkość x nie przyjmuje żadnej konkretnej wartości ! *)
```
rozwiąż równanie

In[104]:=
```
(* Warto też poznać instrukcję Reduce,
które ma za zadanie sprowadzić skomplikowane warunki matematyczne do prostej,
"zredukowanej" postaci *)
```
zredukuj

In[105]:=
```
(* Szukamy wszystkich możliwości
 (a więc zespolonych) rozwiązań tego samego równania *)
```

In[106]:=
```
Reduce[x^3 - x^2 + 2 x + 4 == 0, x]
```
zredukuj

Out[106]=
$x == -1 \,||\, x == 1 - i\sqrt{3} \,||\, x == 1 + i\sqrt{3}$

In[107]:=
```
(* Uwaga: || to logiczne "lub" *)
```

In[108]:=
```
(* Szukamy tylko rzeczywistych rozwiązań tego równania *)
```

In[109]:=
```
Reduce[x^3 - x^2 + 2 x + 4 == 0, x, Reals]
```
zredukuj liczby rze

Out[109]=
$x == -1$

In[110]:=
```
(* Reduce przydaje się zwłaszcza przy nierównościach *)
```
zredukuj

In[111]:=
```
Reduce[x (x^2 - 2) (x^2 - 3) > 0, x, Reals]
```
zredukuj liczby rze

Out[111]=
$-\sqrt{3} < x < -\sqrt{2} \,||\, 0 < x < \sqrt{2} \,||\, x > \sqrt{3}$

In[112]:=
```
Reduce[2^x > E^x, x, Reals]
```
zredukuj liczba E··· liczby rze

Out[112]=
$x < 0$

In[113]:=
```
(* Dopuszczalna jest nierówność jednoczesna *)
```

In[114]:=
```
Simplify[Reduce[2 < Abs[E^(2 * x) - 3] < 3, x, Reals]]
```
uprość zredukuj w··· liczba Eulera liczby rze

Out[114]=
$x < 0 \,||\, \frac{\text{Log}[5]}{2} < x < \frac{\text{Log}[6]}{2}$

In[115]:=
```
(* Inny zapis problemu z logicznym "and" && *)
```

In[116]:=

```
Simplify[Reduce[2 < Abs[E^(2*x) - 3] && Abs[E^(2*x) - 3] < 3, x, Reals]]
```
uprość zredukuj w⋯ liczba Eulera w⋯ liczba Eulera liczby rze

Out[116]=

$$x < 0 \parallel \frac{\text{Log}[5]}{2} < x < \frac{\text{Log}[6]}{2}$$

In[117]:=

```
N[Simplify[Reduce[2 < Abs[E^(2*x) - 3] && Abs[E^(2*x) - 3] < 3, x, Reals]]]
```
· uprość zredukuj w⋯ liczba Eulera w⋯ liczba Eulera liczby rze(

Out[117]=

```
x < 0. || 0.804719 < x < 0.89588
```

In[118]:=

```
(* Dobrze jest wynik sprawdzić na rysunku *)
```

In[119]:=

```
Plot[{2, 3, Abs[E^(2*x) - 3]}, {x, -1, 1}, PlotRange → {1.8, 3.1}]
```
wykres w⋯ liczba Eulera zakres wykresu

Out[119]=



In[120]:=

```
(* Przykład nierówności wymiernej *)
```

In[121]:=

```
Simplify[Reduce[((x + 1)^2 - x^2)/(x*(2*x + 1)) < -1, x, Reals]]
```
uprość zredukuj liczby rze

Out[121]=

```
-1 < x < 0
```

In[122]:=

(* Wyznaczamy "ręcznie" (bo mamy też dedykowaną komendę) dziedzinę funkcji f(x)= $\sqrt{4-x^2}$ +Log[1-x]; w tym celu musimy rozwiązać
logarytm
nierówność z logarytmem i pierwistkiem kwadratowym *)

In[123]:=

```
Simplify[Reduce[4 - x^2 ≥ 0 && 1 - x > 0, x, Reals]]
```
uprość zredukuj liczby rze

Out[123]=

```
-2 ≤ x < 1
```

In[124]:=

```
(* Ale możemy to zrobić po prostu tak: *)
```

In[125]:=

```
FunctionDomain[Sqrt[4 - x ^ 2] + Log[1 - x], x]
```
dziedzina funkcji | pierwiastek kw··· | logarytm

Out[125]=

$-2 \le x < 1$

In[126]:=

```
(* Przykład nierówności trygonometrycznej *)
```

In[127]:=

```
Simplify[Reduce[Sin[x] + Cos[x] > Sqrt[2] * Cos[2 * x], x, Reals]]
```
uprość | zredukuj | sinus | cosinus | pierwia··· | cosinus | liczby rze

Out[127]=

$$c_1 \in \mathbb{Z} \,\&\&\, \left(\pi\left(-\frac{7}{12} + 2\,c_1\right) < x < 2\,\pi\left(-\frac{1}{8} + c_1\right) \,||\, \pi\left(\frac{1}{12} + 2\,c_1\right) < x < 2\,\pi\left(\frac{3}{8} + c_1\right)\right)$$

In[128]:=

```
(* c1 jest dowolną liczbą całkowitą, a litera C jest zastrzeżona dla stałych *)
```
stała

In[129]:=

```
FullSimplify[Reduce[Sin[x] + Cos[x] > Sqrt[2] * Cos[2 * x], x, Reals]]
```
uprość pełniej | zredukuj | sinus | cosinus | pierwia··· | cosinus | liczby rze

Out[129]=

$$c_1 \in \mathbb{Z} \,\&\&\, \left(\pi\left(-\frac{7}{12} + 2\,c_1\right) < x < 2\,\pi\left(-\frac{1}{8} + c_1\right) \,||\, \pi\left(\frac{1}{12} + 2\,c_1\right) < x < 2\,\pi\left(\frac{3}{8} + c_1\right)\right)$$

In[130]:=

```
(* Możemy zawęzić obszar poszukiwania x i otrzymać bardziej jednoznaczny wynik *)
```

In[131]:=

```
FullSimplify[Reduce[Sin[x] + Cos[x] > Sqrt[2] * Cos[2 * x] && 0 < x < 2 * Pi, x, Reals]]
```
uprość pełniej | zredukuj | sinus | cosinus | pierwia··· | cosinus | pi | liczby rze

Out[131]=

$$\frac{\pi}{12} < x < \frac{3\,\pi}{4} \,||\, \frac{17\,\pi}{12} < x < \frac{7\,\pi}{4}$$

In[132]:=

```
(* Możemy szukać rozwiązań w postaci par liczb określonego typu np. całkowitych *)
```

In[133]:=

```
Simplify[Reduce[x ^ 2 + y ^ 2 - 2 * x == 7, {x, y}, Integers]]
```
uprość | zredukuj | zbiór liczb cał

Out[133]=

$(x == -1 \,||\, x == 3) \,\&\&\, (y == -2 \,||\, y == 2)$

In[134]:=

```
(* Przydaje się możliwość rysowanie funkcji y(x) zadanej w postaci uwikłanej *)
```

In[135]:=

```
ContourPlot[Cos[x] + Cos[y] == 1/2, {x, 0, 4 Pi}, {y, 0, 4 Pi}]
```

wykres kontu··· cosinus cosinus pi pi

Out[135]=



In[136]:=

```
(* Możemy podać więcej równań *)
```

In[137]:=

```
ContourPlot[{x^2 + y^2 == 2, x^2 - y^2 == 1/4},
  {x, -2, 2}, {y, -2, 2}, Axes → True, AxesLabel → {"x", "y"}]
```

wykres konturowy · osie · prawda · oznaczenia osi

Out[137]=



In[138]:=

```
(* Rysunek daje wstępne pojęcie o rozwiązaniach, ... *)
```

In[139]:=

```
Reduce[x^2 + y^2 == 2 && x^2 - y^2 == 1/4 && -2 < x < 2 && -2 < y < 2, {x, y}, Reals]
```

zredukuj · liczby rz

Out[139]=

$$\left(x == -\frac{3}{2\sqrt{2}} \; \&\& \; y == -\frac{\sqrt{\frac{7}{2}}}{2}\right) \| \left(x == -\frac{3}{2\sqrt{2}} \; \&\& \; y == \frac{\sqrt{\frac{7}{2}}}{2}\right) \|$$

$$\left(x == \frac{3}{2\sqrt{2}} \; \&\& \; y == -\frac{\sqrt{\frac{7}{2}}}{2}\right) \| \left(x == \frac{3}{2\sqrt{2}} \; \&\& \; y == \frac{\sqrt{\frac{7}{2}}}{2}\right)$$

In[140]:=

```
(* ... ale dokładnego rozwiązania dostarcza Reduce. Są
  to w istocie pary (x,y) stanowiące współrzędne punktów *)
```

zredukuj

In[141]:=

```
N[Reduce[x^2 + y^2 == 2 && x^2 - y^2 == 1/4 && -2 < x < 2 && -2 < y < 2, {x, y}, Reals]]
```

zredukuj · liczby rze

Out[141]=

(x == -1.06066 && y == -0.935414) || (x == -1.06066 && y == 0.935414) ||
(x == 1.06066 && y == -0.935414) || (x == 1.06066 && y == 0.935414)

In[142]:=

```
(* Obszar w płaszczyźnie zadany równaniami *)
```

In[143]:=

```
(* Jaką figurę geometryczną wyznacza na płaszczyźnie
 xy układ nierówności: 2*x+y ≤ 4, (x+4)/2 ≥ y , y ≥ -1 ?  *)
```

In[144]:=

```
RegionPlot[2 * x + y ≤ 4 && (x + 4) / 2 ≥ y && y ≥ -1, {x, -7, 3}, {y, -3 / 2, 3}]
```

wykres regionu na płaszczyźnie

Out[144]=



In[145]:=

```
(* Przekształcanie wykresów zadanej funkcji *)
```

In[146]:=

```
(* Dana jest funkcja f(x)=2*x-3. Narysować wykresy funkcji f(x), f(x)-1,
f(x+1), f(-x),-f(x), -f(-x), |f(x)|, f(2*x), f(1/x), f^(-1)(x),1/f(x) *)
```

In[147]:=

```
Clear[f, x];
```

wyczyść

In[148]:=

```
f[x_] := 2 * x - 3
```

In[149]:=

```
(* Budujemy f^(-1) *)
```

In[150]:=

```
sol = Solve[f[x] == y, x]
```

rozwiąż równanie

Out[150]=

$$\left\{\left\{x \to \frac{3 + y}{2}\right\}\right\}$$

In[151]:=

```
ClearAll[g, y];
```

wyczyść wszystko

In[152]:=

```
g[y_] := (3 + y)/2
```

In[153]:=

```
(* g jest funkcją odwrotną do f, bo zachodzi ... *)
```

In[154]:=

```
g[f[x]]
```

Out[154]=

```
x
```

In[155]:=

```
(* ... oraz  *)
```

In[156]:=

```
f[g[x]]
```

Out[156]=

```
x
```

In[157]:=

```
Plot[{f[x], f[x] - 1, f[x + 1], f[-x], -f[x], -f[-x], Abs[f[x]], f[2 * x], f[1 / x], g[x]},
 {x, -2, 2}, PlotLegends →
  {"f(x)", "f(x)-1", "f(x+1)", "f(-x)", "-f(x)", "-f(-x)", "|f(x)|", "f(2x)", "f(1/x)", "f^-1(x)"}]
```

wykres · wartość bezwzględna · legenda dla grafik

Out[157]=



Granice ciągów i funkcji

In[158]:=

```
(* Przykład "zwykłej" definicji ciągu *)
```

In[159]:=

```
a[n_] := n! / (n ^ 2 + 1)
```

In[160]:=
```
Table[a[n], {n, 1, 12}]
```
tabela

Out[160]=

$$\left\{\frac{1}{2}, \frac{2}{5}, \frac{3}{5}, \frac{24}{17}, \frac{60}{13}, \frac{720}{37}, \frac{504}{5}, \frac{8064}{13}, \frac{181440}{41}, \frac{3628800}{101}, \frac{19958400}{61}, \frac{95800320}{29}\right\}$$

In[161]:=
```
Table[N[a[n]], {n, 1, 12}]
```
tabela przybliżenie numeryczr

Out[161]=

$\{0.5, 0.4, 0.6, 1.41176, 4.61538, 19.4595, 100.8, 620.308, 4425.37, 35928.7, 327187., 3.30346\times 10^6\}$

In[162]:=
```
ClearAll[a]
```
wyczyść wszys

In[163]:=
```
(* Rekurencyjna definicja ciągu Fibonacciego *)
```

In[164]:=
```
a[1] = a[2] = 1; a[n_] := a[n - 2] + a[n - 1];
```

In[165]:=
```
a[3]
```

Out[165]=

2

In[166]:=
```
Table[a[n], {n, 1, 15}]
```
tabela

Out[166]=

{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610}

In[167]:=
```
ClearAll[a];
```
wyczyść wszystk

In[168]:=
```
(* Mathematica potrafi rozpoznać, z jakim ciągiem ma do czynienia *)
```

In[169]:=
```
RSolve[{a[1] == 1, a[2] == 1, a[n] == a[n - 2] + a[n - 1]}, a[n], n]
```
rozwiąż równania rekurencyjne

Out[169]=

{{a[n] → Fibonacci[n]}}

Granice ciągów (przy n→∞)

In[170]:=
```
(* Dzięki programowi Mathematica mamy szansę lepiej zrozumieć definicję ciągu *)
```

In[171]:=
```
(* Możemy sprawdzić wprost z definicji,
że granica poniższych ciągów wynosi zero *)
```

In[172]:=

```
(* Poniżej obrazek z serwisu
   http://www.analizamatematyczna.enhost.pl *)
```



In[173]:=

```
(* przykład nr 1 *)
```

In[174]:=

```
g = 0; Simplify[Reduce[Abs[1 / n - g] < ϵ, n, Reals], n > 0 && ϵ > 0]
```

uprość | zredukuj | wartość bezwzględna | liczby rzeczywiste

Out[174]=

$n\,\epsilon > 1$

In[175]:=

```
(* czyli n > 1/ϵ  *)
```

In[176]:=

```
(* przykład nr 2 *)
```

In[177]:=

```
g = 0; wn = Simplify[Reduce[Abs[1 / (n ^ 2 + n) - g] < ϵ, n, Reals], n > 0 && ϵ > 0 && ϵ < 1]
```

uprość | zredukuj | wartość bezwzględna | liczby rzeczywiste

Out[177]=

$$1 + 2\,n > \sqrt{\frac{4+\epsilon}{\epsilon}}$$

In[178]:=

$$(*\ \text{czyli}\ n > \left(\sqrt{\frac{4+\epsilon}{\epsilon}} - 1\right)/2]\ *)$$

In[179]:=

```
(* przykład nr 3 *)
```

In[180]:=

```
g = 0; Simplify[Reduce[Abs[2 ^ (-n) - g] < ϵ, n, Reals], n > 0 && ϵ > 0 && ϵ < 1]
```

uprość | zredukuj | wartość bezwzględna | liczby rzeczywiste

Out[180]=

$n \operatorname{Log}[2] + \operatorname{Log}[\epsilon] > 0$

In[181]:=

```
(* czyli  n > -Log[ϵ]/Log[2]  *)
```

loga··· | logarytm

In[182]:=

```
(* przykład nr 4 *)
```

In[183]:=

```
g = 3; Simplify[Reduce[Abs[(3*n^2+n-1)/(n^2+5n+2)-g] < ϵ, n, Reals], n > 0 && ϵ > 0 && ϵ < 1]
       uprość   zredukuj wartość bezwzględna                   liczby rzeczywiste
```

Out[183]=

$(5+2\,n)\,\epsilon > 14+\sqrt{196-112\,\epsilon+17\,\epsilon^2}$

In[184]:=

```
(* czyli n > 1/2 (-5+(14+√(196-112 ϵ+17 ϵ^2))/ϵ) *)
```

In[185]:=

```
(* Ilustracja liczbowa, z konkretnym ϵ *)
```

In[186]:=

```
ϵ = 1/1000;
```

In[187]:=

```
soln0 = N[Solve[ϵ (5+2n) == 14 + √(196 - 112 ϵ + 17 ϵ^2), n]];
          rozwiąż równanie
```

In[188]:=

```
n0 = Ceiling[n /. soln0[[1]]];
     sufit
```

In[189]:=

```
Plot[{(3*n^2+n-1)/(n^2+5n+2), 3, 3+ϵ, 3-ϵ},
wykres
 {n, n0-20000, n0+20000}, PlotRange → {2.999-ϵ*11/10, 3+ϵ*11/10}]
                          zakres wykresu
```

Out[189]=



In[190]:=

```
(* Pewne BARDZO znane granice *)
```

In[191]:=

```
Limit[(1+1/n)^n, n → Infinity]
granica                nieskończoność
```

Out[191]=

e

In[192]:=

```
(* Przypominam, że to nie jest "zwykłe" e, ale liczba Eulera,
w programie Mathematica oznaczane przez E albo specjalne, pogrubione e *)
                                                    liczba Eulera
```

In[193]:=

```
Limit[(1 - 1 / n) ^ n, n → Infinity]
```
granica   nieskończon

Out[193]=

$\frac{1}{e}$

In[194]:=

```
(* Wzór Stirlinga *)
```

In[195]:=

```
Limit[n! * E ^ n / (n ^ n * Sqrt[2 * Pi * n]) , n → Infinity]
```
granica   liczba Eulera   pierwia··   pi   nieskończon

Out[195]=

1

In[196]:=

```
(* Z wzoru Stirlinga wynika,
że dla dużych n można n! przybliżyć przez n^n*Sqrt[2*Pi*n]*
  E^(-n). Całkiem nieźle to się zgadza już dla n=50:  *)
N[50 !]
```
pierwi···   pi

liczba Eulera

przybliżenie numeryczne

Out[196]=

3.04141×10^{64}

In[197]:=

```
n = 50; N[n ^ n * Sqrt[2 * Pi * n] * E ^ (-n)]
```
przybl···   pierwia··   pi   liczba Eu

Out[197]=

3.03634×10^{64}

In[198]:=

```
ClearAll[n];
```
wyczyść wszystk

In[199]:=

```
(* Kilka prostych granic ciągów *)
```

In[200]:=

```
Limit[(6 * n ^ 2 - 2 * n + 2) / (3 * n ^ 2 + 5 * n - 2), n → Infinity]
```
granica   nieskończon

Out[200]=

2

In[201]:=

```
Limit[(2 ^ n + 3) / (3 ^ n + 2), n → Infinity]
```
granica   nieskończon

Out[201]=

0

In[202]:=

```
Limit[Sin[n !] / (n + 2), n → Infinity]
```
gran···   sinus   nieskończon

Out[202]=

0

In[203]:=

```
Limit[Sum[2 * i - 1, {i, 1, n}] / (n + 2), n → Infinity]
gran⋯ sumowanie                          nieskończon
```

Out[203]=

∞

In[204]:=

```
Limit[Sum[i, {i, 1, n}] / n ^ 2, n → Infinity]
gran⋯ sumowanie                    nieskończon
```

Out[204]=

$\frac{1}{2}$

In[205]:=

```
Limit[(n ^ 2 + n + 3) / (3 * n ^ 2 - 5 * n - 2), n → Infinity]
granica                                          nieskończon
```

Out[205]=

$\frac{1}{3}$

In[206]:=

```
Limit[Sin[(n ^ 2 + n + 3) * Pi], n → Infinity]
gran⋯ sinus              pi      nieskończon
```

Out[206]=

```
Indeterminate
```

In[207]:=

```
(* Istnieje specjalna komenda do liczenia granicy po liczbach całkowitych ... *)
```

In[208]:=

```
DiscreteLimit[Sin[(n ^ 2 + n + 3) * Pi], n → Infinity]
granica dyskretna sinus          pi      nieskończon
```

Out[208]=

```
0
```

In[209]:=

```
(* ... ale w niektórych przypadkach nawet ona nie pozwala policzyć granicy *)
```

In[210]:=

```
Limit[Sin[n ^ n * Sqrt[2 * Pi * n] * E ^ (-n) * Pi], n → Infinity]
gran⋯ sinus   pierwia⋯ pi    liczba⋯ pi    nieskończon
```

Out[210]=

```
Indeterminate
```

In[211]:=

```
DiscreteLimit[Sin[n ^ n * Sqrt[2 * Pi * n] * E ^ (-n) * Pi], n → Infinity]
granica dyskretna sinus   pierwia⋯ pi    liczba⋯ pi    nieskończon
```

Out[211]=

$$\lim_{\substack{n\to\infty \\ \mathbb{Z}}} \mathrm{Sin}\left[\sqrt{2}\, e^{-n}\, n^{\frac{1}{2}+n}\, \pi^{3/2}\right]$$

In[212]:=

```
Limit[Sin[Sqrt[n ^ 2 + 1] * Pi], n → Infinity]
gran⋯ si⋯ pierwiastek kw⋯ pi    nieskończon
```

Out[212]=

```
Indeterminate
```

In[213]:=

```
DiscreteLimit[Sin[Sqrt[n^2 + 1] * Pi], n → Infinity]
```

granica dyskretna | si··· | pierwiastek kw··· | pi | nieskończonc

Out[213]=

$$\lim_{n \to \infty_{\mathbb{Z}}} \mathrm{Sin}\left[\sqrt{1+n^2}\ \pi\right]$$

Granice funkcji

In[214]:=

```
(* W wersji 12 usunięto wreszcie błąd, który polegał na utożsamianiu
 przez program Mathematica granicy z granicą prawostronną ! *)
```

In[215]:=

```
Plot[Exp[-1/x], {x, -2, 2}, PlotRange → {-2, 10}, AxesLabel → {"x", "f(x)"}]
```

wy··· | funkcja eksponencjalna | zakres wykresu | oznaczenia osi

Out[215]=



In[216]:=

```
Limit[Exp[-1/x], x → 0]
```

gran··· | funkcja eksponencjalı

Out[216]=

```
Indeterminate
```

In[217]:=

```
(* To jest poprawna odpowiedź ! *)
```

In[218]:=

```
Limit[Exp[-1/x], x → 0, Direction → -1]
```

gran··· | funkcja eksponencj··· | kierunek

Out[218]=

```
0
```

In[219]:=

```
(* albo używając bardziej intuicyjnej opcji: *)
```

In[220]:=

```
Limit[Exp[-1/x], x → 0, Direction → "FromAbove"]
```

gran··· | funkcja eksponencj··· | kierunek

Out[220]=

```
0
```

In[221]:=

```
(* oraz lewostronną: *)
```

In[222]:=

```
Limit[Exp[-1 / x], x → 0, Direction → 1]
```
granica | funkcja eksponencj… | kierunek

Out[222]=

∞

In[223]:=

```
(* znowu można użyć bardziej intuicyjnej opcji: *)
```

In[224]:=

```
Limit[Exp[-1 / x], x → 0, Direction → "FromBelow"]
```
granica | funkcja eksponencj… | kierunek

Out[224]=

∞

In[225]:=

```
(* Granice z funkcji wymiernych *)
```

In[226]:=

```
Limit[-5 * x ^ 2 - 3 * x + 1, x → -Infinity]
```
granica | nieskończono

Out[226]=

-∞

In[227]:=

```
Limit[-3 * x ^ 2 * Sqrt[5 * x ^ 2 + 7], x → Infinity]
```
granica | pierwiastek kwadratowy | nieskończono

Out[227]=

-∞

In[228]:=

```
Limit[(-5 * x ^ 5 + 3 * x) / (x ^ 2 - 2), x → -Infinity]
```
granica | nieskończono

Out[228]=

∞

In[229]:=

```
Limit[(x ^ 4 - 1) / (x - 1), x → 1]
```
granica

Out[229]=

4

In[230]:=

```
Limit[(x ^ 2 - 4 * x + 3) / (x - 3), x → 3]
```
granica

Out[230]=

2

In[231]:=

```
Limit[(x ^ 3 + 1) ^ (1 / 4) + (x ^ 4 + 1) ^ (1 / 3), x → Infinity]
```
granica | nieskończono

Out[231]=

∞

In[232]:=
```
Limit[Sin[x] / x, x → 0]
```
gran… sinus

Out[232]=
1

In[233]:=
```
Limit[Cos[x] / (x * Sin[x]), x → 0]
```
gran… cosinus sinus

Out[233]=
∞

In[234]:=
```
Limit[(Sqrt[x] - 1) / (x - 1), x → 1]
```
granica pierwiastek kwadratowy

Out[234]=
$\frac{1}{2}$

In[235]:=
```
Limit[(5 * x ^ 4 - 2 * x ^ 2 - 17) / (8 - 12 * x ^ 3 - x ^ 4), x → Infinity]
```
granica nieskończonc

Out[235]=
-5

In[236]:=
```
Limit[2 / (x - 2) - (x + 6) / (x ^ 2 - 4), x → 2]
```
granica

Out[236]=
$\frac{1}{4}$

Szeregi liczbowe

Zbieżność szeregu liczbowego: czy granica sumy $S_n \equiv \sum_{i=1}^{n} a_i$ istnieje przy n → ∞ ?

In[237]:=
```
(* Szereg harmoniczny jest rozbieżny *)
```

In[238]:=
```
SumConvergence[1 / n, n]
```
zbieżność sumy

Out[238]=
False

In[239]:=
```
(* ale taki naprzemienny szereg jest już zbieżny *)
```

In[240]:=

```
SumConvergence[(-1)^n/n, n]
```
zbieżność sumy

Out[240]=

```
True
```

In[241]:=

```
(* i Mathematica zna sumę tego szeregu: *)
```

In[242]:=

```
Sum[(-1)^n/n, {n, 1, Infinity}]
```
sumowanie nieskończono

Out[242]=

```
-Log[2]
```

In[243]:=

```
SumConvergence[1/n^2, n]
```
zbieżność sumy

Out[243]=

```
True
```

In[244]:=

```
Sum[1/n^2, {n, 1, Infinity}]
```
sumowanie nieskończono

Out[244]=

$$\frac{\pi^2}{6}$$

In[245]:=

```
(* Bardziej ogólny wynik dla szeregu 1/n^alpha: dostaniemy warunek,
kiedy szereg jest zbieżny *)
```

In[246]:=

```
SumConvergence[1/n^alpha, n]
```
zbieżność sumy

Out[246]=

```
Re[alpha] > 1
```

In[247]:=

```
Simplify[Sum[1/n^alpha, {n, 1, Infinity}], Re[alpha] > 1]
```
uprość sumowanie nieskończon··· część rzeczywist

Out[247]=

```
Zeta[alpha]
```

In[248]:=

```
?Zeta
```
funkcja dzeta Riemanna

Out[248]=

Symbol

Zeta[*s*] gives the Riemann zeta function $\zeta(s)$.

Zeta[*s*, *a*] gives the generalized Riemann zeta function $\zeta(s, a)$.

In[249]:=

```
(* Znany wynik dla szeregu geometrycznego *)
```

In[250]:=

```
SumConvergence[q ^ n, n]
```
zbieżność sumy

Out[250]=

```
Abs[q] < 1
```

In[251]:=

```
Simplify[Sum[q ^ n, {n, 0, Infinity}], 0 < Abs[q] < 1]
```
uprość sumowanie nieskończoność wartość bez

Out[251]=

$$\frac{1}{1-q}$$

Pochodne

Pochodna funkcji f(x) w punkcie x0

$$\lim_{\Delta x \to 0} \frac{f(x_0+\Delta x)-f(x_0)}{\Delta x}$$ (https://pl.wikipedia.org/wiki/Pochodna)



In[252]:=

```
(* Używając w programie Mathematica komendy D, musimy zawsze wskazać,
po czym różniczkujemy, nawet jeśli nam wydaje się to zupełnie oczywiste ! *)
```
oblicz pochodną

In[253]:=

```
h = a * x ^ 6
```

Out[253]=

$a\, x^6$

In[254]:=

```
D[h, x]
```
oblicz pochodną

Out[254]=

$6\,a\,x^5$

In[255]:=

```
D[h, a]
```
oblicz pochodną

Out[255]=

x^6

In[256]:=

```
(* pochodne wyższych rzędów *)
```

In[257]:=

```
D[h, {x, 4}]
```
oblicz pochodną

Out[257]=

$360\,a\,x^2$

In[258]:=

```
w[x_] := x ^ 3 / (x ^ 2 + 1)
```

In[259]:=

```
w '[x]
```

Out[259]=

$$-\frac{2\,x^4}{(1+x^2)^2}+\frac{3\,x^2}{1+x^2}$$

In[260]:=

```
w ''[x]
```

Out[260]=

$$\frac{8\,x^5}{(1+x^2)^3}-\frac{14\,x^3}{(1+x^2)^2}+\frac{6\,x}{1+x^2}$$

In[261]:=

```
Simplify[w ''[x]]
```
uprość

Out[261]=

$$-\frac{2\,x\,(-3+x^2)}{(1+x^2)^3}$$

In[262]:=

```
(* i tak dalej ... *)
```

In[263]:=

```
Simplify[w ''''''[x]]
```
uprość

Out[263]=

$$-\frac{720\,x\left(-7+35\,x^2-21\,x^4+x^6\right)}{(1+x^2)^7}$$

In[264]:=
```
Simplify[D[w[x], {x, 6}]]
```
uprość oblicz pochodną

Out[264]=

$$-\frac{720\,x\left(-7+35\,x^2-21\,x^4+x^6\right)}{\left(1+x^2\right)^7}$$

In[265]:=
```
D[Sqrt[2 * x], x]
```
· pierwiastek kwad

Out[265]=

$$\frac{1}{\sqrt{2}\,\sqrt{x}}$$

In[266]:=
```
D[Sin[x] * Cos[x], x]
```
· sinus cosinus

Out[266]=

$$\text{Cos}[x]^2-\text{Sin}[x]^2$$

In[267]:=
```
D[(Sin[2 * x]) ^ 5, x]
```
·· sinus

Out[267]=

$$10\,\text{Cos}[2\,x]\,\text{Sin}[2\,x]^4$$

In[268]:=
```
D[Exp[7 * x + 3], x]
```
· funkcja eksponencja

Out[268]=

$$7\,e^{3+7\,x}$$

In[269]:=
```
D[Tan[x], x]
```
· tangens

Out[269]=

$$\text{Sec}[x]^2$$

In[270]:=
```
(* Sec[x] to 1/Cos[x] *)
```
secans cosinus

In[271]:=
```
Sec[x] === 1 / Cos[x]
```
secans cosinus

Out[271]=
```
True
```

In[272]:=
```
D[Tan[x], x] === 1 / (Cos[x]) ^ 2
```
· tangens cosinus

Out[272]=
```
True
```

In[273]:=

```
D[x^x, x]
```
oblicz pochodną

Out[273]=

$x^x (1 + \text{Log}[x])$

In[274]:=

```
(* Mathematica zna też ogólniejsze twierdzenia o pochodnych,
na przykład o pochodnej iloczynu ... *)
```

In[275]:=

```
D[f1[x] * f2[x], x]
```
oblicz pochodną

Out[275]=

```
f2[x] f1'[x] + f1[x] f2'[x]
```

In[276]:=

```
(* ... lub ilorazu *)
```

In[277]:=

```
Factor[D[f1[x] / f2[x], x]]
```
rozłóż··· oblicz pochodną

Out[277]=

$$\frac{\text{f2}[x]\,\text{f1}'[x] - \text{f1}[x]\,\text{f2}'[x]}{\text{f2}[x]^2}$$

In[278]:=

```
(* Równaniami różniczkowymi będziemy się zajmować bardziej szczegółowo
 na jednym z kolejnych wykładów; teraz sprawdźmy, że funkcja x[t]=
 A*Cos[Sqrt[k/m]*t]+B*Sin[Sqrt[k/m]*t] jest rozwiązaniem równania m*x''[t]+k*x[t]=0 *)
```
co·· pierwiastek kw··· si··· pierwiastek kwadratowy

In[279]:=

```
Clear[x, A, B, k, m]
```
wyczyść

In[280]:=

```
x[t_] := A * Cos[Sqrt[k / m] * t] + B * Sin[Sqrt[k / m] * t]
```
co·· pierwiastek kwadra··· si··· pierwiastek kwad

In[281]:=

```
Simplify[m * x''[t] + k * x[t] == 0]
```
uprość

Out[281]=

```
True
```

In[282]:=

```
(* Weźmy funkcję wielu zmiennych *)
```

In[283]:=

```
ClearAll[x];
```
wyczyść wszystk

In[284]:=

```
v = Sin[x + y] * Cos[x - 2 * z]
```
sinus cosinus

Out[284]=

```
Cos[x - 2 z] Sin[x + y]
```

In[285]:=
```
D[v, x]
```
oblicz pochodną

Out[285]=
```
Cos[x + y] Cos[x - 2 z] - Sin[x + y] Sin[x - 2 z]
```

In[286]:=
```
D[v, y]
```
oblicz pochodną

Out[286]=
```
Cos[x + y] Cos[x - 2 z]
```

In[287]:=
```
D[v, z]
```
oblicz pochodną

Out[287]=
```
2 Sin[x + y] Sin[x - 2 z]
```

In[288]:=
```
(* pochodna po x i y *)
```

In[289]:=
```
D[v, x, y]
```
oblicz pochodną

Out[289]=
```
-Cos[x - 2 z] Sin[x + y] - Cos[x + y] Sin[x - 2 z]
```

In[290]:=
```
(* Rozwinięcie w szereg to bardzo często używane narzędzie,
które pozwala na przybliżanie skomplikowanych funkcji przez wielomiany ! *)
```

In[291]:=
```
Series[Sin[x], {x, 0, 7}]
```
szereg sinus

Out[291]=

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + O[x]^8$$

In[292]:=
```
Normal[Series[Sin[x], {x, 0, 7}]]
```
norma··· szereg sinus

Out[292]=

$$x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040}$$

In[293]:=

```
Plot[{Sin[x], x - x^3/6 + x^5/120 - x^7/5040}, {x, 0, 3}]
```

wyk··· sinus

Out[293]=



In[294]:=

```
Series[Cos[x], {x, 0, 7}]
```

szereg cosinus

Out[294]=

$$1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720} + O[x]^8$$

In[295]:=

```
Normal[Series[Cos[x], {x, 0, 7}]]
```

norma··· szereg cosinus

Out[295]=

$$1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720}$$

In[296]:=

```
Plot[{Cos[x], 1 - x^2/2 + x^4/24 - x^6/720}, {x, 0, 3}]
```

wyk··· cosinus

Out[296]=



Jedno z wielu zastosowań pochodnych: równania stycznych do wykresu funkcji

In[297]:=
```
(* Styczna do paraboli y=x^2  o równaniu  y = a*x + b *)
```

In[298]:=
```
Clear[f, x];
```
wyczyść

In[299]:=
```
f[x_] = x ^ 2
```

Out[299]=

x^2

In[300]:=
```
df = D[f[x], x]
```
oblicz poch

Out[300]=

2 x

In[301]:=
```
x0 = 3;
```

In[302]:=
```
(* Współczynnik kierunkowy prostej jest równy pochodnej w punkce x0 *)
```

In[303]:=
```
a = df /. x → x0
```

Out[303]=

6

In[304]:=
```
(* Współczynnik b dostajemy z warunku,
ze prosta musi przechodzić przez punkt (x0,f(x0)) *)
```

In[305]:=
```
solb = Solve[a * x0 + b == f[x0], b]
```
rozwiąż równanie

Out[305]=

{{b → -9}}

In[306]:=
```
b = b /. solb[[1]]
```

Out[306]=

-9

In[307]:=

```
Plot[{f[x], a*x+b, f[x0]}, {x, x0-3, x0+3}, PlotRange→All, AxesLabel→{"x", "x^2"}]
```
wykres zakres wykresu ws··· oznaczenia osi

Out[307]=



In[308]:=

```
ClearAll[a, b, x0];
```
wyczyść wszystko

Badanie przebiegu zmienności funkcji

In[309]:=

```
(* Zbadać przebieg zmienności funkcji f(x)=(x^2+2)/(x+1) *)
```

In[310]:=

```
ClearAll[f, x, y];
```
wyczyść wszystko

In[311]:=

```
f[x_] = (x^2+2)/(x+1)
```

Out[311]=

$$\frac{2+x^2}{1+x}$$

In[312]:=

```
(* Dziedzina funkcji jest w danym przypadku oczywista ... *)
```

In[313]:=

```
Simplify[Reduce[1+x≠0, x, Reals]]
```
uprość zredukuj liczby rze

Out[313]=

$1+x\neq 0$

In[314]:=

```
(* ... ale niekiedy specjalna instrukcja FunctionDomain oszczędza sporo pracy *)
```
dziedzina funkcji

In[315]:=

```
Df = FunctionDomain[f[x], x, Reals]
      dziedzina funkcji          liczby rz
```

Out[315]=

$x < -1 \| x > -1$

In[316]:=

```
(* Przeciwdziedzina (zbiór wartości) funkcji: uwaga na składnię ! *)
```

In[317]:=

```
Rf = FunctionRange[f[x], x, y]
      zakres funkcji
```

Out[317]=

$y \le -2 - 2\sqrt{3} \| y \ge -2 + 2\sqrt{3}$

In[318]:=

```
(* To już nie było takie oczywiste !  *)
```

In[319]:=

```
(* Granice na krańcach dziedziny,
a w tym przypadku w minus i plus nieskończoności *)
```

In[320]:=

```
g1 = Limit[f[x], x → -Infinity]
      granica         nieskończon
```

Out[320]=

$-\infty$

In[321]:=

```
g2 = Limit[f[x], x → Infinity]
      granica        nieskończon
```

Out[321]=

∞

In[322]:=

```
(* Funkcja nie ma więc asymptot poziomych *)
```

In[323]:=

```
(* Granica lewostronna i prawostronna przy x→ -1 *)
```

In[324]:=

```
g3 = Limit[f[x], x → -1, Direction → 1]
      granica              kierunek
```

Out[324]=

$-\infty$

In[325]:=

```
g4 = Limit[f[x], x → -1, Direction → -1]
      granica              kierunek
```

Out[325]=

∞

In[326]:=

```
(* W danym przypadku odpowiedź jest prosta, ale warto wiedzieć,
jak sprawdzić, czy funkcja jest parzysta ... *)
```

In[327]:=
```
FullSimplify[ForAll[x, Df, f[x] == f[-x]]]
```
uprość pełniej kwantyfikator ogólny

Out[327]=
```
False
```

In[328]:=
```
(* ... lub czy funkcja jest nieparzysta ! *)
```

In[329]:=
```
FullSimplify[ForAll[x, Df, f[-x] == -f[x]]]
```
uprość pełniej kwantyfikator ogólny

Out[329]=
```
False
```

In[330]:=
```
(* pierwsza pochodna *)
```

In[331]:=
```
df1 = Simplify[D[f[x], x]]
```
uprość oblicz pochc

Out[331]=

$$\frac{-2 + 2x + x^2}{(1 + x)^2}$$

In[332]:=
```
(* Dla jakich x pochodna jest ujemna ? *)
```

In[333]:=
```
FullSimplify[Reduce[df1 < 0, x, Reals]]
```
uprość pełniej zredukuj liczby rze

Out[333]=

$\left(1 + \sqrt{3} + x > 0 \,\&\&\, 1 + x < 0\right) \,||\, -1 < x < -1 + \sqrt{3}$

In[334]:=

(* Funkcja jest malejąca dla $-1-\sqrt{3} < x < -1$ oraz
gdy $-1 < x < -1+\sqrt{3}$ *)

In[335]:=
```
(* Dla jakich x pochodna jest dodatnia ? *)
```

In[336]:=
```
FullSimplify[Reduce[df1 > 0, x, Reals]]
```
uprość pełniej zredukuj liczby rze

Out[336]=

$1 + \sqrt{3} + x < 0 \,||\, 1 + x > \sqrt{3}$

In[337]:=

(* Funkcja jest rosnąca dla $x < -1-\sqrt{3}$ oraz dla $x > \sqrt{3}-1$ *)

In[338]:=
```
(* Dla jakich x pochodna przyjmuje wartość zero ? *)
```

In[339]:=
```
sdf1zero = Solve[df1 == 0, x]
```
rozwiąż równanie

Out[339]=

$\left\{\left\{x \to -1 - \sqrt{3}\right\}, \left\{x \to -1 + \sqrt{3}\right\}\right\}$

In[340]:=
```
(* Lista punktów x, dla których pierwsza pochodna przyjmuje wartość zero *)
```

In[341]:=
```
zera1poch = x /. sdf1zero
```

Out[341]=
$\{-1-\sqrt{3}, -1+\sqrt{3}\}$

In[342]:=
```
N[zera1poch = x /. sdf1zero]
```
przybliżenie numeryczne

Out[342]=
```
{-2.73205, 0.732051}
```

In[343]:=
```
(* Po uproszczeniu wartości funkcji w tych punktach *)
```

In[344]:=
```
FullSimplify[f[zera1poch]]
```
uprość pełniej

Out[344]=
$\{-2(1+\sqrt{3}), 2(-1+\sqrt{3})\}$

In[345]:=
```
N[FullSimplify[f[zera1poch]]]
```
· uprość pełniej

Out[345]=
```
{-5.4641, 1.4641}
```

In[346]:=
```
df2 = Simplify[D[df1, x]]
```
uprość oblicz pocho

Out[346]=
$$\frac{6}{(1+x)^3}$$

In[347]:=
```
(* Dla jakich x druga pochodna jest ujemna ? *)
```

In[348]:=
```
FullSimplify[Reduce[df2 < 0, x, Reals]]
```
uprość pełniej zredukuj liczby rze

Out[348]=
$x < -1$

In[349]:=
```
(* Funkcja jest wklęsła dla x < -1 *)
```

In[350]:=
```
(* Dla jakich x druga pochodna jest dodatnia ? *)
```

In[351]:=
```
FullSimplify[Reduce[df2 > 0, x, Reals]]
```
uprość pełniej zredukuj liczby rze

Out[351]=
$x > -1$

In[352]:=
```
(* Funkcja jest wypukła dla x > -1 *)
```

In[353]:=

```
(* Dla jakich x druga pochodna przyjmuje wartość zero ? *)
```

In[354]:=

```
sdf2zero = Solve[df2 == 0, x]
```

rozwiąż równanie

Out[354]=

```
{}
```

In[355]:=

```
(* Funkcja nie ma punktów przegięcia *)
```

In[356]:=

```
(* Wartości drugiej pochodnej dla miejsc zerowych pierwszej pochodnej *)
```

In[357]:=

```
zera1poch
```

Out[357]=

$\{-1-\sqrt{3}, -1+\sqrt{3}\}$

In[358]:=

```
wartosci2poch = df2 /. sdf1zero
```

Out[358]=

$\left\{-\frac{2}{\sqrt{3}}, \frac{2}{\sqrt{3}}\right\}$

In[359]:=

(* Dlatego f(x) ma w punkcie $\left(-1-\sqrt{3}, -2\left(1+\sqrt{3}\right)\right)$ maksimum lokalne,
a w punkcie $\left(-1+\sqrt{3}, 2\left(-1+\sqrt{3}\right)\right)$ minimum lokalne *)

In[360]:=

```
(* Na koniec sprawdzimy jeszcze, czy istnieją asymptoty ukośne *)
```

In[361]:=

```
a1 = Limit[f[x] / x, x → -Infinity]
```

granica nieskończonc

Out[361]=

1

In[362]:=

```
b1 = Limit[f[x] - a1 * x, x → -Infinity]
```

granica nieskończonc

Out[362]=

-1

In[363]:=

```
a2 = Limit[f[x] / x, x → Infinity]
```

granica nieskończonc

Out[363]=

1

In[364]:=

```
b2 = Limit[f[x] - a2 * x, x → Infinity]
```

granica nieskończonc

Out[364]=

-1

In[365]:=

```
(* Prosta o równaniu y=x-1 jest asymptotą ukośną zarówno dla x→ -∞,
jak i dla x→ ∞ *)
```

In[366]:=

```
(* Wykres funkcji wraz z asymptotami *)
```

In[367]:=

```
wykres = Plot[{f[x], a1 * x + b1}, {x, -4, 3}, AxesLabel → {"x", "f(x)"}]
```

wykres · oznaczenia osi

Out[367]=



In[368]:=

```
(* Asymptotę pionową musimy narysować przy pomocy innej instrukcji *)
```

In[369]:=

```
pion = ParametricPlot[{-1, t}, {t, -20, 15}, PlotStyle → {Dashed, Green}];
```

wykres parametryczny · styl grafiki · linia prz··· · zielony

In[370]:=

```
Show[wykres, pion]
```

pokaż

Out[370]=



Całka nieoznaczona

In[371]:=

```
(* To operacja odwrotna do liczenia pochodnej *)
```

In[372]:=

```
a =.;
```

In[373]:=

```
Integrate[x ^ a, x]
```

całka

Out[373]=

$$\frac{x^{1+a}}{1+a}$$

In[374]:=

```
(* Wynik jest poprawny, jeśli a ≠ -1 *)
```

In[375]:=

$$D\left[\frac{x^{1+a}}{1+a}, x\right]$$

oblicz pochodną

Out[375]=

$$x^a$$

In[376]:=

```
Integrate[1 / x, x]
```

całka

Out[376]=

```
Log[x]
```

In[377]:=

```
D[Log[x], x]
```

· logarytm

Out[377]=

$$\frac{1}{x}$$

In[378]:=

```
Integrate[Sqrt[(x + 7) / (x + 2)], x]
```

całka pierwiastek kwadratowy

Out[378]=

$$\frac{\sqrt{\frac{7+x}{2+x}}\left((2+x)\sqrt{7+x}+5\sqrt{2+x}\,\text{ArcTanh}\left[\frac{1}{\sqrt{\frac{2+x}{7+x}}}\right]\right)}{\sqrt{7+x}}$$

In[379]:=

$$\text{Simplify}\left[D\left[\frac{1}{\sqrt{7+x}}\sqrt{\frac{7+x}{2+x}}\left((2+x)\sqrt{7+x}+5\sqrt{2+x}\,\text{ArcSinh}\left[\frac{\sqrt{2+x}}{\sqrt{5}}\right]\right), x\right]\right]$$

uprość oblicz pochodną arcus sinus hiperboliczny

Out[379]=

$$\sqrt{\frac{7+x}{2+x}}$$

In[380]:=

```
Integrate[Cos[x], x]
```

całka cosinus

Out[380]=

```
Sin[x]
```

In[381]:=

```
Integrate[Sin[x], x]
```
całka sinus

Out[381]=

$-\mathrm{Cos}[x]$

In[382]:=

```
Integrate[Tan[x], x]
```
całka tangens

Out[382]=

$-\mathrm{Log}[\mathrm{Cos}[x]]$

In[383]:=

```
Integrate[Cot[x], x]
```
całka cotangens

Out[383]=

$\mathrm{Log}[\mathrm{Sin}[x]]$

In[384]:=

```
Integrate[Sin[x] * Cos[x] / ((Sin[x]) ^ 4 + (Cos[x]) ^ 2), x]
```
całka sinus cosinus sinus cosinus

Out[384]=

$$-\frac{\mathrm{ArcTan}\left[\frac{1-2\,\mathrm{Tan}[x]}{\sqrt{3}}\right]+\mathrm{ArcTan}\left[\frac{1+2\,\mathrm{Tan}[x]}{\sqrt{3}}\right]}{\sqrt{3}}$$

In[385]:=

```
Integrate[Exp[x], x]
```
całka funkcja eksp

Out[385]=

e^{x}

In[386]:=

```
Integrate[x ^ 3 / (x ^ 2 + 1), x]
```
całka

Out[386]=

$$\frac{x^2}{2}-\frac{1}{2}\,\mathrm{Log}\left[1+x^2\right]$$

In[387]:=

```
Integrate[Sqrt[1 + x ^ 2], x]
```
całka pierwiastek kwadrato

Out[387]=

$$\frac{1}{2}\left(x\,\sqrt{1+x^2}+\mathrm{ArcTanh}\left[\frac{x}{\sqrt{1+x^2}}\right]\right)$$

In[388]:=

```
Integrate[Sqrt[1 - x ^ 2], x]
```
całka pierwiastek kwadrato

Out[388]=

$$\frac{1}{2}\left(x\,\sqrt{1-x^2}+\mathrm{ArcTan}\left[\frac{x}{\sqrt{1-x^2}}\right]\right)$$

In[389]:=

```
Integrate[Sqrt[x ^ 2 - 1], x]
```
całka pierwiastek kwadratow

Out[389]=

$$\frac{1}{2} x \sqrt{-1+x^2} - \frac{1}{2} \text{ArcTanh}\left[\frac{x}{\sqrt{-1+x^2}}\right]$$

In[390]:=

```
Integrate[1 / Sqrt[1 + x ^ 2], x]
```
całka pierwiastek kwadratow

Out[390]=

$$\text{ArcTanh}\left[\frac{x}{\sqrt{1+x^2}}\right]$$

In[391]:=

```
(* Można oczywiście używać bardziej eleganckiego zapisu: *)
```

In[392]:=

$$\int \text{ArcSinh}[x] \, dx$$
arcus sinus hiperb

Out[392]=

$$-\sqrt{1+x^2} + x \, \text{ArcSinh}[x]$$

In[393]:=

```
Integrate[1 / Sqrt[1 - x ^ 2], x]
```
całka pierwiastek kwadratow

Out[393]=

$$\text{ArcTan}\left[\frac{x}{\sqrt{1-x^2}}\right]$$

In[394]:=

```
Integrate[1 / Sqrt[x ^ 2 - 1], x]
```
całka pierwiastek kwadratow

Out[394]=

$$\text{ArcTanh}\left[\frac{x}{\sqrt{-1+x^2}}\right]$$

In[395]:=

$$\int \text{Log}\left[x + \sqrt{-1+x^2}\right] dx$$
logarytm

Out[395]=

$$-\sqrt{-1+x^2} + x \, \text{Log}\left[x + \sqrt{-1+x^2}\right]$$

In[396]:=

$$\int \left(-\sqrt{-1+x^2} + x \, \text{Log}\left[x + \sqrt{-1+x^2}\right]\right) dx$$
logarytm

Out[396]=

$$-\frac{3}{4} x \sqrt{-1+x^2} + \frac{1}{4} \text{ArcTanh}\left[\frac{x}{\sqrt{-1+x^2}}\right] + \frac{1}{2} x^2 \, \text{Log}\left[x + \sqrt{-1+x^2}\right]$$

In[397]:=

$$\int\left(-\frac{3}{4}\,x\,\sqrt{-1+x^2}+\frac{1}{4}\,\text{Log}\left[x+\sqrt{-1+x^2}\,\right]+\frac{1}{2}\,x^2\,\text{Log}\left[x+\sqrt{-1+x^2}\,\right]\right)dx$$

logarytm logarytm

Out[397]=

$$\frac{1}{4}\left(\left(-\frac{13}{9}-\frac{2\,x^2}{9}\right)\sqrt{-1+x^2}-\left(-1+x^2\right)^{3/2}+\frac{1}{3}\,x\left(3+2\,x^2\right)\text{Log}\left[x+\sqrt{-1+x^2}\,\right]\right)$$

In[398]:=

```
(* Czasem trzeba zachować ostrożność,
bo Mathematica lubi używać zespolonego rachunku,
a my oczekujemy funkcji rzeczywistej *)
```

In[399]:=

```
Integrate[Cos[x] / (1 + (Sin[x]) ^ 2 + (Cos[x]) ^ 3), x]
```

całka cosinus sinus cosinus

Out[399]=

$$\frac{1}{25}\left(\left(1+7\,i\right)\sqrt{1-2\,i}\,\text{ArcTan}\left[\sqrt{1-2\,i}\,\text{Tan}\left[\frac{x}{2}\right]\right]+\left(1-7\,i\right)\sqrt{1+2\,i}\,\text{ArcTan}\left[\sqrt{1+2\,i}\,\text{Tan}\left[\frac{x}{2}\right]\right]-5\,\text{Tan}\left[\frac{x}{2}\right]\right)$$

In[400]:=

```
(* Rozwiązanie całki nieoznaczonej w postaci funkcji rzeczywistej
 wymaga sporo zachodu i jest przedstawione w notatniku calka_z_NOF_2.nb,
który jest dostępny na mojej stronie *)
```

In[401]:=

```
(* Oczywiście nie powinniśmy żądać rzeczy
 niemożliwych: niektóre całki nie wyrażają się przez funkcje elementarne *)
```

In[402]:=

```
Integrate[Exp[-x ^ 2], x]
```

całka funkcja eksponen

Out[402]=

$$\frac{1}{2}\,\sqrt{\pi}\,\text{Erf}[x]$$

In[403]:=

$$\int e^{-x^2}\,dx$$

Out[403]=

$$\frac{1}{2}\,\sqrt{\pi}\,\text{Erf}[x]$$

In[404]:=

$$\partial_x\left(\frac{1}{2}\,\sqrt{\pi}\,\text{Erf}[x]\right)$$

funkcja b

Out[404]=

$$e^{-x^2}$$

In[405]:=

$$\int\frac{\text{Cos}[x]}{x}\,dx$$

Out[405]=

```
CosIntegral[x]
```

In[406]:=

$$\int \frac{\text{Sin}[x]}{x} \, dx$$

Out[406]=

```
SinIntegral[x]
```

Całka oznaczona

Całka oznaczona w przypadku funkcji jednej zmiennej $\int_a^b f(x)\,dx$

(rysunek ze strony
http://blog.etrapez.pl/calki-oznaczone/calki-oznaczone-definicja/)



Jeśli F'(x) = f(x), to $\int_a^b f(x)\,dx = F(b) - F(a)$

In[407]:=

```
Plot[x, {x, 1, 3}, PlotRange → {0, 3}, AxesLabel → {"x", "x"}, Filling → Axis]
```
wykres | zakres wykresu | oznaczenia osi | wypełnienie | oś

Out[407]=



In[408]:=

```
PoleTrapezu = (1 / 2) * (3 + 1) * (3 - 1)
```

Out[408]=

4

In[409]:=

```
Integrate[x, x]
```
całka

Out[409]=

$$\frac{x^2}{2}$$

In[410]:=

$$\text{CalkaOznaczona} = \left(\frac{x^2}{2} \text{ /. } x \to 3\right) - \left(\frac{x^2}{2} \text{ /. } x \to 1\right)$$

Out[410]=

4

In[411]:=

```
Plot[Sin[x], {x, 0, Pi}, AxesLabel → {"x", "sin(x)"}, Filling → Axis]
```
wy··· | sinus | pi | oznaczenia osi | wypełnienie | oś

Out[411]=



In[412]:=

```
Integrate[Sin[x], {x, 0, Pi}]
```

całka sinus pi

Out[412]=

2

WAŻNA UWAGA PRAKTYCZNA:
W wielu przypadkach opłaca się policzyć najpierw całkę nieoznaczoną, a następnie przy jej pomocy całkę oznaczoną !

In[413]:=

```
(* Poniższa instrukcja byłaby wykonywana bardzo dłuuuugo: *)
```

In[414]:=

$$(* \; \mathrm{Simplify}\left[\int_0^{L/v1} \sqrt{v1^2+\left(alpha+\frac{beta\ t\ v1}{L}\right)^2}\, dt, v1>0\&\&alpha>0\&\&beta>0\&\&L>0\right] \; *)$$

uprość

In[415]:=

```
(* Dlatego zrobimy tak: *)
```

In[416]:=

$$s0 = \mathrm{Simplify}\left[\int \sqrt{v1^2 + \left(alpha + \frac{beta\ t\ v1}{L}\right)^2}\, dt,\ v1 > 0\ \&\&\ alpha > 0\ \&\&\ beta > 0\ \&\&\ L > 0\right]$$

uprość

Out[416]=

$$\frac{1}{2\ beta\ v1}\left((alpha\ L + beta\ t\ v1)\sqrt{alpha^2 + \frac{2\ alpha\ beta\ t\ v1}{L} + \left(1 + \frac{beta^2\ t^2}{L^2}\right) v1^2} - 2\ L\ v1^2\ \mathrm{ArcTanh}\left[\frac{L\left(\sqrt{alpha^2 + v1^2} - \sqrt{alpha^2 + \frac{2\ alpha\ beta\ t\ v1}{L} + \left(1 + \frac{beta^2\ t^2}{L^2}\right) v1^2}\right)}{beta\ t\ v1}\right]\right)$$

In[447]:=

```
u = Simplify[(s0 /. t → L / v1)];
```

uprość

```
(* Nie można bezpośrednio podstawić t=0 i musimy sobie poradzić, licząc granicę! *)
```

In[448]:=

```
l = Simplify[Limit[s0, t → 0]]
```

uprość granica

Out[448]=

$$\frac{L\left(alpha\ \sqrt{alpha^2 + v1^2} + 2\ v1^2\ \mathrm{ArcTanh}\left[\frac{alpha}{\sqrt{alpha^2+v1^2}}\right]\right)}{2\ beta\ v1}$$

In[417]:=

```
(* A następnie: *)
```

In[450]:=

```
s1 = FullSimplify[u - l]
```
uprość pełniej

Out[450]=

$$\frac{1}{2\ \text{beta}\ \text{v1}} \text{L} \left(-\text{alpha}\sqrt{\text{alpha}^2+\text{v1}^2}+\text{alpha}\sqrt{(\text{alpha}+\text{beta})^2+\text{v1}^2}+\text{beta}\sqrt{(\text{alpha}+\text{beta})^2+\text{v1}^2}-2\ \text{v1}^2\left(\text{ArcTanh}\left[\frac{\text{alpha}}{\sqrt{\text{alpha}^2+\text{v1}^2}}\right]+\text{ArcTanh}\left[\frac{\sqrt{\text{alpha}^2+\text{v1}^2}-\sqrt{(\text{alpha}+\text{beta})^2+\text{v1}^2}}{\text{beta}}\right]\right)\right)$$

In[419]:=

In[420]:=

```
(* Pole koła o promieniu r *)
```

In[421]:=

```
r = 2;
RegionPlot[x^2 + y^2 ≤ r^2, {x, -r, r}, {y, -r, r}, Axes → True, AxesLabel → {"x", "y"}]
```
wykres regionu na płaszczyźnie | osie | prawda | oznaczenia osi

Out[421]=



In[422]:=

```
(* To samo przy pomocy innej instrukcji i innych opcji *)
```

In[423]:=

```
r = 2;
Plot[{-Sqrt[r^2 - x^2], Sqrt[r^2 - x^2]}, {x, -r, r},
 AspectRatio → 1, Frame → True, Filling → {1 → {2}}, AxesLabel → {"x", "y"}]
```

wykres | pierwiastek kwadra… | pierwiastek kwadratowy | format obrazu | ramka | prawda | wypełnienie | oznaczenia osi

Out[423]=



In[424]:=

```
r =.; Simplify[2 * Integrate[Sqrt[r^2 - x^2], {x, -r, r}], r > 0]
```

uprość | całka | pierwiastek kwadratowy

Out[424]=

πr^2

In[425]:=

```
(* Kilka badzo ważnych całek niewłaściwych *)
```

In[426]:=

```
Plot[Exp[-x^2], {x, -4, 4}, PlotRange → All, Filling → Axis]
```

wy… | funkcja eksponencjalna | zakres wykresu | ws… | wypełnienie | oś

Out[426]=



In[427]:=

```
Integrate[Exp[-x^2], {x, -Infinity, Infinity}]
```

całka | funkcja eksponen… | nieskończ… | nieskończono

Out[427]=

$\sqrt{\pi}$

In[428]:=

```
a = 2; Plot[Exp[-a*x], {x, 0, 4}, PlotRange → All, Filling → Axis]
```

wy… | funkcja eksponencjalna | zakres wykresu | ws… | wypełnienie | oś

Out[428]=



In[429]:=

```
a =.; Simplify[Integrate[Exp[-a*x], {x, 0, Infinity}], a > 0]
```

uprość | całka | funkcja eksponencja… | nieskończoność

Out[429]=

$\frac{1}{a}$

In[430]:=

```
f1[x_] = Max[Sin[x] / x, 0];
```

… | sinus

In[431]:=

```
f2[x_] = Min[Sin[x] / x, 0];
```

mi… | sinus

In[432]:=

```
Plot[{f1[x], f2[x]}, {x, -15, 15}, PlotRange → All,
wykres                                zakres wykresu wszyst
  Filling → Axis, AxesLabel → {"x", "sin(x)/x"}]
  wypełnienie  oś     oznaczenia osi
```

Out[432]=



In[433]:=

```
Integrate[Sin[x] / x, {x, -Infinity, Infinity}]
całka        sinus             nieskończ···  nieskończono
```

Out[433]=

π

In[434]:=

```
Clear[f1, f2];
wyczyść
```

In[435]:=

```
f1[x_] = Max[Sin[x ^ 2], 0];
         ···  sinus
```

In[436]:=

```
f2[x_] = Min[Sin[x ^ 2], 0];
         mi··  sinus
```

In[437]:=

```
Plot[{f1[x], f2[x]}, {x, -4, 4}, PlotRange → All,
wykres                               zakres wykresu  wszystk
 Filling → Axis, AxesLabel → {"x", "sin(x^2)"}]
 wypełnienie  oś    oznaczenia osi
```

Out[437]=



In[438]:=

```
Integrate[Sin[x^2], {x, -Infinity, Infinity}]
całka     sinus            nieskończ···  nieskończono
```

Out[438]=

$$\sqrt{\frac{\pi}{2}}$$

In[439]:=

```
Integrate[Cos[x^2], {x, -Infinity, Infinity}]
całka     cosinus          nieskończ···  nieskończono
```

Out[439]=

$$\sqrt{\frac{\pi}{2}}$$

In[440]:=

```
(* Objętość pod wykresem funkcji dwóch
 zmiennych wymaga podwójnej całki oznaczonej  *)
```

In[441]:=

```
Plot3D[x^2 + 2*y, {x, 0, 3}, {y, 0, 5}, AxesLabel → {"x", "y", "f(x,y)"}]
```
wykres trójwymiarowy　　oznaczenia osi

Out[441]=



In[442]:=

```
Integrate[x^2 + 2*y, {x, 0, 3}, {y, 0, 5}]
```
całka

Out[442]=

120

In[443]:=

```
(* Policzmy teraz objętość kuli, czyli podwojoną objętość półkuli o promieniu r,
używając współrzednych kartezjańskich *)
```

In[444]:=

```
Clear[r];
```
wyczyść

```
2*Integrate[Sqrt[r^2 - x^2 - y^2], {x, -r, r}, {y, -Sqrt[r^2 - x^2], Sqrt[r^2 - x^2]}]
```
całka　pierwiastek kwadratowy　pierwiastek kwadra···　pierwiastek kwadratov

Out[444]=

$$\frac{4\pi r^3}{3}$$

In[445]:=

```
(* Kula jest kostką we współrzędnych sferycznych i trójwymiarowa
 całka oznaczona (po rp, theta i phi) ma bardzo proste granice *)
```

In[446]:=

```
Integrate[rp^2*Sin[theta], {rp, 0, r}, {theta, 0, Pi}, {phi, 0, 2*Pi}]
```
całka　sinus　pi　pi

Out[446]=

$$\frac{4\pi r^3}{3}$$

Inne zastosowania całek oznaczonych w fizyce i technice poznacie Państwo niebawem !